
package dp32_types is

constant unit_delay : Time := 1 ns;

type bool_to_bit_table is array (boolean) of bit;
constant bool_to_bit : bool_to_bit_table;

subtype bit_32 is bit_vector(31 downto 0);
type bit_32_array is array (integer range <>) of bit_32;
function resolve_bit_32 (driver : in bit_32_array) return bit_32;
subtype bus_bit_32 is resolve_bit_32 bit_32;

subtype bit_8 is bit_vector(7 downto 0);

subtype CC_bits is bit_vector(2 downto 0);
subtype cm_bits is bit_vector(3 downto 0);

constant op_add : bit_8 := X"00";
constant op_sub : bit_8 := X"01";
constant op_mul : bit_8 := X"02";
constant op_div : bit_8 := X"03";
constant op_addq : bit_8 := X"10";
constant op_subq : bit_8 := X"11";
constant op_mulq : bit_8 := X"12";
constant op_divq : bit_8 := X"13";
constant op_land : bit_8 := X"04";
constant op_lor : bit_8 := X"05";
constant op_lxor : bit_8 := X"06";
constant op_lmask : bit_8 := X"07";
constant op_ld : bit_8 := X"20";
constant op_st : bit_8 := X"21";
constant op_ldq : bit_8 := X"30";
constant op_stq : bit_8 := X"31";
constant op_br : bit_8 := X"40";
constant op_brq : bit_8 := X"50";
constant op_bi : bit_8 := X"41";
constant op_biq : bit_8 := X"51";

function bits_to_int (bits : in bit_vector) return integer;
function bits_to_natural (bits : in bit_vector) return natural;
procedure int_to_bits (int : in integer; bits : out bit_vector);

end dp32_types;

package body dp32_types is

constant bool_to_bit : bool_to_bit_table :=
(false => '0', true => '1');

function resolve_bit_32 (driver : in bit_32_array) return bit_32 is

constant float_value : bit_32 := X"0000_0000";
variable result : bit_32 := float_value;

begin
for i in driver'range loop

result := result or driver(i);
end loop;
return result;

end resolve_bit_32;

function bits_to_int (bits : in bit_vector) return integer is

variable temp : bit_vector(bits'range);
variable result : integer := 0;

begin
if bits(bits'left) = '1' then -- negative number

temp := not bits;
else

temp := bits;
end if;
for index in bits'range loop -- sign bit of temp = '0'

result := result * 2 + bit'pos(temp(index));
end loop;
if bits(bits'left) = '1' then

result := (-result) - 1;
end if;
return result;

end bits_to_int;

function bits_to_natural (bits : in bit_vector) return natural is

variable result : natural := 0;

begin
for index in bits'range loop

result := result * 2 + bit'pos(bits(index));
end loop;
return result;

end bits_to_natural;

procedure int_to_bits (int : in integer; bits : out bit_vector) is

variable temp : integer;
variable result : bit_vector(bits'range);

begin
if int < 0 then

temp := -(int+1);
else

temp := int;
end if;
for index in bits'reverse_range loop

result(index) := bit'val(temp rem 2);
temp := temp / 2;

end loop;
if int < 0 then

result := not result;
result(bits'left) := '1';

end if;
bits := result;

end int_to_bits;

end dp32_types;

� � � � � � � � � á } � ~ � � ~ � � � â �

 ! ( 1 8 + $ ' " * +
bits_to_int

$ * + : ( ) ' # % 9 " ' : ( $ ' * ) ) ( & ) ( # ( + ' " + ; % ' > * # �$ * . & / " . ( + ' # " ; + ( - " + ' ( ; ( ) " + ' * % + " + ' ( ; ( ) ' 2 & ( : % / 8 ( 7  ! ( / * $ % / : % ) " % 9 / (
temp

" # - ( $ / % ) ( - ' * 9 ( % 9 " ' : ( $ ' * ) * 1 ' ! ( # % . ( # " [ ( % + - " + - ( , ) % + ; ( % # ' ! (
& % ) % . ( ' ( ) 9 " ' # 7  ! ( : % ) " % 9 / ( ) ( # 8 / ' " # " + " ' " % / " # ( - ' * [ ( ) * > ! ( + ' ! ( 1 8 + $ ' " * +
" # " + : * = ( - 0 % + - # 8 9 # ( r 8 ( + ' / 2 8 # ( - ' * % $ $ 8 . 8 / % ' ( ' ! ( > ( " ; ! ' ( - 9 " ' : % / 8 ( # " +



_ �   ¡ ¢ £ ¤ d ¥ j ¦ d ¤ § ¨ b c d g © ª a © « j ¬ d § § j « _ ` ã

use work.dp32_types.all;

entity dp32 is

generic (Tpd : Time := unit_delay);

port (d_bus : inout bus_bit_32 bus;
a_bus : out bit_32;
read, write : out bit;
fetch : out bit;
ready : in bit;
phi1, phi2 : in bit;
reset : in bit);

end dp32;

use work.dp32_types.all;

architecture behaviour of dp32 is

subtype reg_addr is natural range 0 to 255;
type reg_array is array (reg_addr) of bit_32;

begin -- behaviour of dp32

process

variable reg : reg_array;
variable PC : bit_32;
variable current_instr : bit_32;
variable op: bit_8;
variable r3, r1, r2 : reg_addr;
variable i8 : integer;
alias cm_i : bit is current_instr(19);
alias cm_V : bit is current_instr(18);
alias cm_N : bit is current_instr(17);
alias cm_Z : bit is current_instr(16);
variable cc_V, cc_N, cc_Z : bit;
variable temp_V, temp_N, temp_Z : bit;
variable displacement, effective_addr : bit_32;

procedure memory_read (addr : in bit_32;
fetch_cycle : in boolean;
result : out bit_32) is

begin
-- start bus cycle with address output
a_bus <= addr after Tpd;
fetch <= bool_to_bit(fetch_cycle) after Tpd;
wait until phi1 = '1';
if reset = '1' then

return;
end if;
--
-- T1 phase
--
read <= '1' after Tpd;
wait until phi1 = '1';
if reset = '1' then

return;
end if;
--
-- T2 phase
--
loop

wait until phi2 = '0';
if reset = '1' then

return;
end if;
-- end of T2
if ready = '1' then

result := d_bus;
exit;

end if;
end loop;
wait until phi1 = '1';
if reset = '1' then

return;
end if;
--
-- Ti phase at end of cycle
--
read <= '0' after Tpd;

end memory_read;

procedure memory_write (addr : in bit_32;
data : in bit_32) is

begin
-- start bus cycle with address output
a_bus <= addr after Tpd;
fetch <= '0' after Tpd;
wait until phi1 = '1';
if reset = '1' then

return;
end if;
--
-- T1 phase
--
write <= '1' after Tpd;
wait until phi2 = '1';
d_bus <= data after Tpd;
wait until phi1 = '1';
if reset = '1' then

return;
end if;
--
-- T2 phase
--
loop

wait until phi2 = '0';
if reset = '1' then

return;
end if;
-- end of T2
exit when ready = '1';

end loop;
wait until phi1 = '1';
if reset = '1' then

return;
end if;
--
-- Ti phase at end of cycle
--
write <= '0' after Tpd;
d_bus <= null after Tpd;

end memory_write;

procedure add (result : inout bit_32;
op1, op2 : in integer;
V, N, Z : out bit) is

begin
if op2 > 0 and op1 > integer'high-op2 then -- positive overflow

int_to_bits(((integer'low+op1)+op2)-integer'high-1, result);
V := '1';

elsif op2 < 0 and op1 < integer'low-op2 then -- negative overflow
int_to_bits(((integer'high+op1)+op2)-integer'low+1, result);
V := '1';

else
int_to_bits(op1 + op2, result);
V := '0';

end if;
N := result(31);
Z := bool_to_bit(result = X"0000_0000");

end add;

procedure subtract (result : inout bit_32;
op1, op2 : in integer;
V, N, Z : out bit) is

begin
if op2 < 0 and op1 > integer'high+op2 then -- positive overflow

int_to_bits(((integer'low+op1)-op2)-integer'high-1, result);
V := '1';

elsif op2 > 0 and op1 < integer'low+op2 then -- negative overflow
int_to_bits(((integer'high+op1)-op2)-integer'low+1, result);
V := '1';

else
int_to_bits(op1 - op2, result);
V := '0';

end if;
N := result(31);
Z := bool_to_bit(result = X"0000_0000");

end subtract;

procedure multiply (result : inout bit_32;
op1, op2 : in integer;
V, N, Z : out bit) is

begin
if ((op1>0 and op2>0) or (op1<0 and op2<0)) -- result positive

and (abs op1 > integer'high / abs op2) then -- positive overflow
int_to_bits(integer'high, result);
V := '1';

elsif ((op1>0 and op2<0) or (op1<0 and op2>0)) -- result negative
and ((- abs op1) < integer'low / abs op2) then -- negative overflow

int_to_bits(integer'low, result);
V := '1';

else
int_to_bits(op1 * op2, result);
V := '0';

end if;
N := result(31);
Z := bool_to_bit(result = X"0000_0000");

end multiply;

procedure divide (result : inout bit_32;
op1, op2 : in integer;
V, N, Z : out bit) is

begin
if op2=0 then

if op1>=0 then  -- positive overflow
int_to_bits(integer'high, result);

else
int_to_bits(integer'low, result);

end if;
V := '1';

else
int_to_bits(op1 / op2, result);
V := '0';

end if;
N := result(31);
Z := bool_to_bit(result = X"0000_0000");

end divide;

begin
--
-- check for reset active
--
if reset = '1' then

read <= '0' after Tpd;
write <= '0' after Tpd;
fetch <= '0' after Tpd;
d_bus <= null after Tpd;
PC := X"0000_0000";
wait until reset = '0';

end if;
--
-- fetch next instruction
--
memory_read(PC, true, current_instr);
if reset /= '1' then

add(PC, bits_to_int(PC), 1, temp_V, temp_N, temp_Z);
--
-- decode & execute
--
op := current_instr(31 downto 24);
r3 := bits_to_natural(current_instr(23 downto 16));
r1 := bits_to_natural(current_instr(15 downto 8));
r2 := bits_to_natural(current_instr(7 downto 0));
i8 := bits_to_int(current_instr(7 downto 0));

case op is
when op_add =>

add(reg(r3), bits_to_int(reg(r1)), bits_to_int(reg(r2)),
cc_V, cc_N, cc_Z);

when op_addq =>
add(reg(r3), bits_to_int(reg(r1)), i8, cc_V, cc_N, cc_Z);

when op_sub =>
subtract(reg(r3), bits_to_int(reg(r1)), bits_to_int(reg(r2)),

cc_V, cc_N, cc_Z);
when op_subq =>

subtract(reg(r3), bits_to_int(reg(r1)), i8, cc_V, cc_N, cc_Z);
when op_mul =>

multiply(reg(r3), bits_to_int(reg(r1)), bits_to_int(reg(r2)),
cc_V, cc_N, cc_Z);

when op_mulq =>
multiply(reg(r3), bits_to_int(reg(r1)), i8, cc_V, cc_N, cc_Z);

when op_div =>
divide(reg(r3), bits_to_int(reg(r1)), bits_to_int(reg(r2)),

cc_V, cc_N, cc_Z);
when op_divq =>

divide(reg(r3), bits_to_int(reg(r1)), i8, cc_V, cc_N, cc_Z);
when op_land =>

reg(r3) := reg(r1) and reg(r2);
cc_Z := bool_to_bit(reg(r3) = X"0000_0000");

when op_lor =>
reg(r3) := reg(r1) or reg(r2);
cc_Z := bool_to_bit(reg(r3) = X"0000_0000");

when op_lxor =>
reg(r3) := reg(r1) xor reg(r2);
cc_Z := bool_to_bit(reg(r3) = X"0000_0000");

when op_lmask =>
reg(r3) := reg(r1) and not reg(r2);
cc_Z := bool_to_bit(reg(r3) = X"0000_0000");

when op_ld =>
memory_read(PC, true, displacement);
if reset /= '1' then

add(PC, bits_to_int(PC), 1, temp_V, temp_N, temp_Z);
add(effective_addr,

bits_to_int(reg(r1)), bits_to_int(displacement),
temp_V, temp_N, temp_Z);

memory_read(effective_addr, false, reg(r3));
end if;

when op_ldq =>
add(effective_addr,

bits_to_int(reg(r1)), i8,
temp_V, temp_N, temp_Z);

memory_read(effective_addr, false, reg(r3));
when op_st =>

memory_read(PC, true, displacement);
if reset /= '1' then

add(PC, bits_to_int(PC), 1, temp_V, temp_N, temp_Z);
add(effective_addr,

bits_to_int(reg(r1)), bits_to_int(displacement),
temp_V, temp_N, temp_Z);

memory_write(effective_addr, reg(r3));
end if;

when op_stq =>
add(effective_addr,

bits_to_int(reg(r1)), i8,
temp_V, temp_N, temp_Z);

memory_write(effective_addr, reg(r3));
when op_br =>

memory_read(PC, true, displacement);
if reset /= '1' then

add(PC, bits_to_int(PC), 1, temp_V, temp_N, temp_Z);
add(effective_addr,

bits_to_int(PC), bits_to_int(displacement),
temp_V, temp_N, temp_Z);

if ((cm_V and cc_V) or (cm_N and cc_N) or (cm_Z and cc_Z))
= cm_i then

PC := effective_addr;
end if;

end if;
when op_bi =>

memory_read(PC, true, displacement);
if reset /= '1' then

add(PC, bits_to_int(PC), 1, temp_V, temp_N, temp_Z);
add(effective_addr,

bits_to_int(reg(r1)), bits_to_int(displacement),
temp_V, temp_N, temp_Z);

if ((cm_V and cc_V) or (cm_N and cc_N) or (cm_Z and cc_Z))
= cm_i then

PC := effective_addr;
end if;

end if;
when op_brq =>

add(effective_addr,
bits_to_int(PC), i8,
temp_V, temp_N, temp_Z);

if ((cm_V and cc_V) or (cm_N and cc_N) or (cm_Z and cc_Z))
= cm_i then

PC := effective_addr;
end if;

when op_biq =>
add(effective_addr,

bits_to_int(reg(r1)), i8,
temp_V, temp_N, temp_Z);

if ((cm_V and cc_V) or (cm_N and cc_N) or (cm_Z and cc_Z))
= cm_i then

PC := effective_addr;
end if;

when others =>
assert false report "illegal instruction" severity warning;

end case;
end if; -- reset /= '1'

end process;

end behaviour;

use work.dp32_types.all;

entity clock_gen is
generic (Tpw : Time; -- clock pulse width

 Tps : Time); -- pulse separation between phases
port (phi1, phi2 : out bit;

reset : out bit);
end clock_gen;

architecture behaviour of clock_gen is

constant clock_period : Time := 2*(Tpw+Tps);

begin

reset_driver :
reset <= '1', '0' after 2*clock_period+Tpw;

clock_driver : process
begin

phi1 <=  '1', '0' after Tpw;
phi2 <= '1' after Tpw+Tps, '0' after Tpw+Tps+Tpw;
wait for clock_period;

end process clock_driver;

end behaviour;

� � � � � � � � � � � � � � } � � È � � � ~ � Ë } z � } É Õ � � ~ � � � ¯ � � �

@ A ñ A Ø M E F Í M D I O
Z + ( > % 2 * 1 ' ( # ' " + ; ' ! ( 9 ( ! % : " * 8 ) % / . * - ( / * 1 ' ! ( 3 4 5 6 & ) * $ ( # # * ) " # ' *

$ * + + ( $ ' " ' " + % ' ( # ' 9 ( + $ ! $ " ) $ 8 " ' 0 # ! * > + " + Q " ; 8 ) ( � � � V 7  ! (
clock_gen$ * . & * + ( + ' ; ( + ( ) % ' ( # ' ! ( ' > * � & ! % # ( $ / * $ = % + - ' ! ( ) ( # ( ' # " ; + % / ' * - ) " : ( ' ! (

& ) * $ ( # # * ) 7  ! ( . ( . * ) 2 # ' * ) ( # % ' ( # ' & ) * ; ) % . % + - - % ' % 7 Þ ( > ) " ' (
9 ( ! % : " * 8 ) % / . * - ( / # 1 * ) ' ! ( # ( ' > * $ * . & * + ( + ' # 0 % + - $ * + + ( $ ' ' ! ( . " + %

# ' ) 8 $ ' 8 ) % / - ( # $ ) " & ' " * + * 1 ' ! ( ' ( # ' 9 ( + $ ! 7
Q " ; 8 ) ( � � � � / " # ' # ' ! ( ( + ' " ' 2 - ( $ / % ) % ' " * + % + - 9 ( ! % : " * 8 ) % / % ) $ ! " ' ( $ ' 8 ) ( * 1

' ! ( $ / * $ = ; ( + ( ) % ' * ) 7  ! (
clock_gen

( + ' " ' 2 ! % # ' > * 1 * ) . % / ; ( + ( ) " $ $ * + # ' % + ' # 7
Tpw

" # ' ! ( & 8 / # ( > " - ' ! 1 * ) ( % $ ! * 1
phi1

% + -
phi2

0 ' ! % ' " # 0 ' ! ( ' " . ( 1 * ) > ! " $ !
( % $ ! $ / * $ = " # ß � ß 7

Tps
" # ' ! ( & 8 / # ( # ( & % ) % ' " * + 0 ' ! % ' " # 0 ' ! ( ' " . ( 9 ( ' > ( ( + * + (

$ / * $ = # " ; + % / $ ! % + ; " + ; ' * ß V ß % + - ' ! ( * ' ! ( ) $ / * $ = # " ; + % / $ ! % + ; " + ; ' * ß � ß 7



_ �   ¡ ¢ £ ¤ d ¥ j ¦ d ¤ § ¨ b c d g © ª a © « j ¬ d § § j « _ ` ë ã

\ % # ( - * + ' ! ( # ( : % / 8 ( # 0 ' ! ( $ / * $ = & ( ) " * - " # ' > " $ ( ' ! ( # 8 . * 1 ' ! ( & 8 / # ( > " - ' !
% + - ' ! ( # ( & % ) % ' " * + 7

 ! ( % ) $ ! " ' ( $ ' 8 ) ( * 1 ' ! ( $ / * $ = ; ( + ( ) % ' * ) $ * + # " # ' # * 1 ' > * $ * + $ 8 ) ) ( + '
# ' % ' ( . ( + ' # 0 * + ( ' * - ) " : ( ' ! (

reset
# " ; + % / % + - ' ! ( * ' ! ( ) ' * - ) " : ( ' ! ( $ / * $ =

# " ; + % / # 7  ! ( ) ( # ( ' - ) " : ( ) # $ ! ( - 8 / ( # % ß � ß : % / 8 ( * +
reset

> ! ( + " ' " # % $ ' " : % ' ( -
% ' # " . 8 / % ' " * + " + " ' " % / " # % ' " * + 0 1 * / / * > ( - 9 2 % ß V ß % / " ' ' / ( % 1 ' ( ) ' > * $ / * $ = & ( ) " * - #

/ % ' ( ) 7  ! " # $ * + $ 8 ) ) ( + ' # ' % ' ( . ( + ' " # + ( : ( ) # 8 9 # ( r 8 ( + ' / 2 ) ( % $ ' " : % ' ( - 0 # " + $ (
" ' # > % : ( 1 * ) . / " # ' - * ( # + * ' ) ( 1 ( ) ' * % + 2 # " ; + % / # 7  ! ( $ / * $ = - ) " : ( ) & ) * $ ( # # 0
> ! ( + % $ ' " : % ' ( - 0 # $ ! ( - 8 / ( # % & 8 / # ( * +

phi1
" . . ( - " % ' ( / 2 0 1 * / / * > ( - 9 2 % & 8 / # (

* +
phi2

0 % + - ' ! ( + # 8 # & ( + - # 1 * ) % $ / * $ = & ( ) " * - 7 Þ ! ( + " ' ) ( # 8 . ( # 0 " ' ) ( & ( % ' # 0
# $ ! ( - 8 / " + ; ' ! ( + ( , ' $ / * $ = $ 2 $ / ( 7

 ! ( ( + ' " ' 2 - ( $ / % ) % ' " * + % + - 9 ( ! % : " * 8 ) % / % ) $ ! " ' ( $ ' 8 ) ( * 1 ' ! ( . ( . * ) 2
. * - 8 / ( % ) ( # ! * > + " + Q " ; 8 ) ( � � � 6 7  ! ( % ) $ ! " ' ( $ ' 8 ) ( 9 * - 2 $ * + # " # ' # * 1 * + (
& ) * $ ( # # ' * " . & / ( . ( + ' ' ! ( 9 ( ! % : " * 8 ) 7  ! ( & ) * $ ( # # $ * + ' % " + # % + % ) ) % 2
: % ) " % 9 / ( ' * ) ( & ) ( # ( + ' ' ! ( # ' * ) % ; ( * 1 ' ! ( . ( . * ) 2 7 Þ ! ( + ' ! ( & ) * $ ( # # " #

% $ ' " : % ' ( - 0 " ' & / % $ ( # ' ! ( * 8 ' & 8 ' & * ) ' # " + % + " + " ' " % / # ' % ' ( � ' ! ( - % ' % 9 8 #
- " # $ * + + ( $ ' ( - % + - ' ! (

ready
9 " ' + ( ; % ' ( - 7 P ' ' ! ( + > % " ' # 1 * ) ( " ' ! ( ) % ) ( % - * )

> ) " ' ( $ * . . % + - 7 Þ ! ( + * + ( * 1 ' ! ( # ( * $ $ 8 ) # 0 ' ! ( % - - ) ( # # " # # % . & / ( - % + -
$ * + : ( ) ' ( - 1 ) * . % 9 " ' � : ( $ ' * ) ' * % + 8 . 9 ( ) 7 P 1 " ' " # > " ' ! " + ' ! ( % - - ) ( # # 9 * 8 + - #
* 1 ' ! ( . ( . * ) 2 0 ' ! ( $ * . . % + - " # % $ ' ( - 8 & * + 7

Q * ) % > ) " ' ( $ * . . % + - 0 ' ! (
ready

9 " ' " # % # # ( ) ' ( - % 1 ' ( ) % - ( / % 2 ) ( & ) ( # ( + ' " + ;' ! ( > ) " ' ( % $ $ ( # # ' " . ( * 1 ' ! ( . ( . * ) 2 0 % + - ' ! ( + ' ! ( . * - ( / > % " ' # 8 + ' " / ' ! ( ( + -
* 1 ' ! ( > ) " ' ( $ 2 $ / ( 7 < ' ' ! % ' ' " . ( 0 ' ! ( : % / 8 ( * + ' ! ( - % ' % 9 8 # 1 ) * . %
& ) * & % ; % ' " * + - ( / % 2 9 ( 1 * ) ( ! % + - " # # % . & / ( - % + - > ) " ' ' ( + " + ' * ' ! ( . ( . * ) 2

% ) ) % 2 7  ! ( 8 # ( * 1 ' ! " # - ( / % 2 ( - : % / 8 ( . * - ( / # ' ! ( 1 % $ ' ' ! % ' . ( . * ) 2 - ( : " $ ( #
% $ ' 8 % / / 2 # ' * ) ( ' ! ( - % ' % ' ! % ' > % # : % / " - % # ( ' 8 & � ' " . ( 9 ( 1 * ) ( ' ! ( ' ) " ; ; ( ) " + ; ( - ; (

* 1 ' ! ( $ * . . % + - 9 " ' 7
Q * ) % ) ( % - $ * . . % + - 0 ' ! ( - % ' % 1 ) * . ' ! ( . ( . * ) 2 % ) ) % 2 " # % $ $ ( # # ( - % + -

& / % $ ( - * + ' ! ( - % ' % 9 8 # % 1 ' ( ) % - ( / % 2 7  ! " # - ( / % 2 ) ( & ) ( # ( + ' # ' ! ( ) ( % - % $ $ ( # #
' " . ( * 1 ' ! ( . ( . * ) 2 7  ! (

ready
9 " ' " # % / # * % # # ( ) ' ( - % 1 ' ( ) ' ! ( - ( / % 2 0 " + - " $ % ' " + ;' ! % ' ' ! ( & ) * $ ( # # * ) . % 2 $ * + ' " + 8 ( 7  ! ( . ( . * ) 2 ' ! ( + > % " ' # 8 + ' " / ' ! ( ( + - * 1

' ! ( ) ( % - $ 2 $ / ( 7
< ' ' ! ( ( + - * 1 % . ( . * ) 2 $ 2 $ / ( 0 ' ! ( & ) * $ ( # # ) ( & ( % ' # 0 # ( ' ' " + ; ' ! ( - % ' % 9 8 #

% + -
ready

9 " ' - ) " : ( ) # ' * ' ! ( " ) " + " ' " % / # ' % ' ( 0 % + - > % " ' " + ; 1 * ) ' ! ( + ( , '
$ * . . % + - 7

Q " ; 8 ) ( � � � 5 # ! * > # ' ! ( ( + ' " ' 2 - ( $ / % ) % ' " * + % + - # ' ) 8 $ ' 8 ) % / % ) $ ! " ' ( $ ' 8 ) ( * 1
' ! ( ' ( # ' 9 ( + $ ! $ " ) $ 8 " ' 7  ! ( ( + ' " ' 2 $ * + ' % " + # + * & * ) ' # 0 # " + $ ( ' ! ( ) ( % ) ( + *
( , ' ( ) + % / $ * + + ( $ ' " * + # ' * ' ! ( ' ( # ' 9 ( + $ ! 7  ! ( % ) $ ! " ' ( $ ' 8 ) ( 9 * - 2 $ * + ' % " + #
$ * . & * + ( + ' - ( $ / % ) % ' " * + # 1 * ) ' ! ( $ / * $ = - ) " : ( ) 0 ' ! ( . ( . * ) 2 % + - ' ! ( & ) * $ ( # # * ) 7
 ! ( & * ) ' # " + ' ! ( # ( $ * . & * + ( + ' - ( $ / % ) % ' " * + # $ * ) ) ( # & * + - ( , % $ ' / 2 ' * ' ! * # ( * 1 ' ! (
( + ' " ' 2 - ( $ / % ) % ' " * + # 7  ! ( ) ( % ) ( + * 1 * ) . % / ; ( + ( ) " $ $ * + # ' % + ' # 0 # * ' ! ( % $ ' 8 % / #
1 * ) ' ! ( ; ( + ( ) " $ # " + ' ! ( ( + ' " ' 2 - ( $ / % ) % ' " * + # > " / / 9 ( # & ( $ " 1 " ( - " + %
$ * + 1 " ; 8 ) % ' " * + 7  ! ( % ) $ ! " ' ( $ ' 8 ) ( 9 * - 2 + ( , ' - ( $ / % ) ( # ' ! ( # " ; + % / # > ! " $ ! % ) (
8 # ( - ' * $ * + + ( $ ' ' ! ( $ * . & * + ( + ' # ' * ; ( ' ! ( ) 7  ! ( # ( # " ; + % / # . % 2 9 ( ' ) % $ ( - 9 2 %

# " . 8 / % ' " * + . * + " ' * ) > ! ( + ' ! ( # " . 8 / % ' " * + " # ) 8 + 7  ! ( $ * + $ 8 ) ) ( + '
# ' % ' ( . ( + ' # * 1 ' ! ( % ) $ ! " ' ( $ ' 8 ) ( 9 * - 2 $ * + # " # ' * 1 ' ! ( ' ! ) ( ( $ * . & * + ( + '

" + # ' % + $ ( # 7



_ ` a ì b c d e f g h i j j k l j j k

use work.dp32_types.all;

entity memory is
generic (Tpd : Time := unit_delay);
port (d_bus : inout bus_bit_32 bus;

a_bus : in bit_32;
read, write : in bit;
ready : out bit);

end memory;

architecture behaviour of memory is
begin

process

constant low_address : integer := 0;
constant high_address : integer := 65535;
type memory_array is

array (integer range low_address to high_address) of bit_32;
variable mem : memory_array;
variable address : integer;

begin
--
-- put d_bus and reply into initial state
--
d_bus <= null after Tpd;
ready <= '0' after Tpd;
--
-- wait for a command
--
wait until (read = '1') or (write = '1');
--
-- dispatch read or write cycle
--
address := bits_to_int(a_bus);
if address >= low_address and address <= high_address then

-- address match for this memory
if write = '1' then

ready <= '1' after Tpd;
wait until write = '0'; -- wait until end of write cycle
mem(address) := d_bus'delayed(Tpd); -- sample data from Tpd ago

else -- read = '1'
d_bus <= mem(address) after Tpd; -- fetch data
ready <= '1' after Tpd;
wait until read = '0'; -- hold for read cycle

end if;
end if;

end process;

end behaviour;

use work.dp32_types.all;

entity dp32_test is
end dp32_test;

architecture structure of dp32_test is

component clock_gen
port (phi1, phi2 : out bit;

reset : out bit);
end component;

component dp32
port (d_bus : inout bus_bit_32 bus;

a_bus : out bit_32;
read, write : out bit;
fetch : out bit;
ready : in bit;
phi1, phi2 : in bit;
reset : in bit);

end component;

component memory
port (d_bus : inout bus_bit_32 bus;

a_bus : in bit_32;
read, write : in bit;
ready : out bit);

end component;

signal d_bus : bus_bit_32 bus;
signal a_bus : bit_32;
signal read, write : bit;
signal fetch : bit;
signal ready : bit;
signal phi1, phi2 : bit;
signal reset : bit;

begin

cg : clock_gen
port map (phi1 => phi1, phi2 => phi2, reset => reset);

proc : dp32
port map (d_bus => d_bus, a_bus => a_bus,

read => read, write => write, fetch => fetch,
ready => ready,
phi1 => phi1, phi2 => phi2, reset => reset);

mem : memory
port map (d_bus => d_bus, a_bus => a_bus,

read => read, write => write, ready => ready);

end structure;

configuration dp32_behaviour_test of dp32_test is
for structure

for cg : clock_gen
use entity work.clock_gen(behaviour)

generic map (Tpw => 8 ns, Tps => 2 ns);
end for;
for mem : memory

use entity work.memory(behaviour);
end for;
for proc : dp32

use entity work.dp32(behaviour);
end for;

end for;
end dp32_behaviour_test;

 1.                                include dp32.inc $

 2.

 3.                                !!! conventions:

 4.                                !!! r0 = 0

 5.                                !!! r1 scratch

 6.

 7.                                begin

 8.   (     0) [07000000         ]  initr0

 9.                                start:

10.   (     1) [10020000         ]  addq(r2, r0, 0) ! r2 := 0

11.                                loop:

12.   (     2) [21020000 00000008]  sta(r2, counter) ! counter := r2

13.   (     4) [10020201         ]  addq(r2, r2, 1) ! increment r2

14.   (     5) [1101020A         ]  subq(r1, r2, 10) ! if r2 = 10 then

15.   (     6) [500900FA         ]  brzq(start) !   restart

16.   (     7) [500000FA         ]  braq(loop) ! else next loop

17.

18.                                counter:

19.   (     8) [00000000         ]  data(0)

20.                                end

use work.dp32_types.all;

entity mux2 is
generic (width : positive;

Tpd : Time := unit_delay);
port (i0, i1 : in bit_vector(width-1 downto 0);

y : out bit_vector(width-1 downto 0);
sel : in bit);

end mux2;

architecture behaviour of mux2 is
begin

with sel select
y <= i0 after Tpd when '0',

i1 after Tpd when '1';
end behaviour;

use work.dp32_types.all;

entity latch is
generic (width : positive;

Tpd : Time := unit_delay);
port (d : in bit_vector(width-1 downto 0);

q : out bit_vector(width-1 downto 0);
en : in bit);

end latch;

architecture behaviour of latch is

begin

process (d, en)
begin

if en = '1' then
q <= d after Tpd;

end if;
end process;

end behaviour;

use work.dp32_types.all;

entity buffer_32 is
generic (Tpd : Time := unit_delay);
port (a : in bit_32;

b : out bus_bit_32 bus;
en : in bit);

end buffer_32;

architecture behaviour of buffer_32 is

begin

b_driver: process (en, a)
begin

if en = '1' then
b <= a after Tpd;

else
b <= null after Tpd;

end if;
end process b_driver;

end behaviour;

use work.dp32_types.all;

entity signext_8_32 is
generic (Tpd : Time := unit_delay);
port (a : in bit_8;

b : out bus_bit_32 bus;
en : in bit);

end signext_8_32;

architecture behaviour of signext_8_32 is

begin

b_driver: process (en, a)
begin

if en = '1' then
b(7 downto 0) <= a after Tpd;
if a(7) = '1' then

b(31 downto 8) <= X"FFFF_FF" after Tpd;
else

b(31 downto 8) <= X"0000_00" after Tpd;
end if;

else
b <= null after Tpd;

end if;
end process b_driver;

end behaviour;

use work.dp32_types.all;

entity latch_buffer_32 is
generic (Tpd : Time := unit_delay);
port (d : in bit_32;

q : out bus_bit_32 bus;
latch_en : in bit;
out_en : in bit);

end latch_buffer_32;

architecture behaviour of latch_buffer_32 is

begin

process (d, latch_en, out_en)
variable latched_value : bit_32;

begin
if latch_en = '1' then

latched_value := d;
end if;
if out_en = '1' then

q <= latched_value after Tpd;
else

q <= null after Tpd;
end if;

end process;

end behaviour;

use work.dp32_types.all;

entity PC_reg is
generic (Tpd : Time := unit_delay);
port (d : in bit_32;

q : out bus_bit_32 bus;
latch_en : in bit;
out_en : in bit;
reset : in bit);

end PC_reg;

architecture behaviour of PC_reg is

begin

process (d, latch_en, out_en, reset)
variable master_PC, slave_PC : bit_32;

begin
if reset = '1' then

slave_PC := X"0000_0000";
elsif latch_en = '1' then

master_PC := d;
else

slave_PC := master_PC;
end if;
if out_en = '1' then

q <= slave_PC after Tpd;
else

q <= null after Tpd;
end if;

end process;

end behaviour;

use work.dp32_types.all;

entity reg_file_32_rrw is
generic (depth : positive; -- number of address bits

Tpd : Time := unit_delay;
Tac : Time := unit_delay);

port (a1 : in bit_vector(depth-1 downto 0);
q1 : out bus_bit_32 bus;
en1 : in bit;
a2 : in bit_vector(depth-1 downto 0);
q2 : out bus_bit_32 bus;
en2 : in bit;
a3 : in bit_vector(depth-1 downto 0);
d3 : in bit_32;
en3 : in bit);

end reg_file_32_rrw;

architecture behaviour of reg_file_32_rrw is

begin

reg_file: process (a1, en1, a2, en2, a3, d3, en3)

subtype reg_addr is natural range 0 to depth-1;
type register_array is array (reg_addr) of bit_32;

variable registers : register_array;

begin
if en3 = '1' then

registers(bits_to_natural(a3)) := d3;
end if;
if en1 = '1' then

q1 <= registers(bits_to_natural(a1)) after Tac;
else

q1 <= null after Tpd;
end if;
if en2 = '1' then

q2 <= registers(bits_to_natural(a2)) after Tac;
else

q2 <= null after Tpd;
end if;

end process reg_file;

end behaviour;

package ALU_32_types is

type ALU_command is (disable, pass1, incr1,
add, subtract, multiply, divide,
log_and, log_or, log_xor, log_mask);

end ALU_32_types;

use work.dp32_types.all, work.ALU_32_types.all;

entity ALU_32 is
generic (Tpd : Time := unit_delay);
port (operand1 : in bit_32;

operand2 : in bit_32;
result : out bus_bit_32 bus;
cond_code : out CC_bits;
command : in ALU_command);

end ALU_32;

architecture behaviour of ALU_32 is

alias cc_V : bit is cond_code(2);
alias cc_N : bit is cond_code(1);
alias cc_Z : bit is cond_code(0);

begin

ALU_function: process (operand1, operand2, command)

variable a, b : integer;
variable temp_result : bit_32;

begin
case command is

when add | subtract | multiply | divide =>
a := bits_to_int(operand1);
b := bits_to_int(operand2);

when incr1 =>
a := bits_to_int(operand1);
b := 1;

when others =>
null;

end case;
case command is

when disable =>
null;

when pass1 =>
temp_result := operand1;

when log_and =>
 temp_result := operand1 and operand2;

when log_or =>
temp_result := operand1 or operand2;

when log_xor =>
temp_result := operand1 xor operand2;

when log_mask =>
temp_result := operand1 and not operand2;

when add | incr1 =>
if b > 0 and a > integer'high-b then -- positive overflow

int_to_bits(((integer'low+a)+b)-integer'high-1, temp_result);
cc_V <= '1' after Tpd;

elsif b < 0 and a < integer'low-b then -- negative overflow
 int_to_bits(((integer'high+a)+b)-integer'low+1, temp_result);
cc_V <= '1' after Tpd;

else
int_to_bits(a + b, temp_result);
cc_V <= '0' after Tpd;

end if;
when subtract =>

if b < 0 and a > integer'high+b then -- positive overflow
int_to_bits(((integer'low+a)-b)-integer'high-1, temp_result);
cc_V <= '1' after Tpd;

elsif b > 0 and a < integer'low+b then -- negative overflow
int_to_bits(((integer'high+a)-b)-integer'low+1, temp_result);
cc_V <= '1' after Tpd;

else
int_to_bits(a - b, temp_result);
cc_V <= '0' after Tpd;

end if;

when multiply =>
if ((a>0 and b>0) or (a<0 and b<0)) -- result positive

and (abs a > integer'high / abs b) then
-- positive overflow
int_to_bits(integer'high, temp_result);
cc_V <= '1' after Tpd;

elsif ((a>0 and b<0) or (a<0 and b>0)) -- result negative
and ((- abs a) < integer'low / abs b) then

-- negative overflow
int_to_bits(integer'low, temp_result);
cc_V <= '1' after Tpd;

else
int_to_bits(a * b, temp_result);
cc_V <= '0' after Tpd;

end if;
when divide =>

if b=0 then
if a>=0 then -- positive overflow

int_to_bits(integer'high, temp_result);
else

int_to_bits(integer'low, temp_result);
end if;
cc_V <= '1' after Tpd;

else
int_to_bits(a / b, temp_result);
cc_V <= '0' after Tpd;

end if;
end case;
if command /= disable then

result <= temp_result after Tpd;
else

result <= null after Tpd;
end if;
cc_Z <= bool_to_bit(temp_result = X"00000000") after Tpd;
cc_N <= bool_to_bit(temp_result(31) = '1') after Tpd;

end process ALU_function;

end behaviour;

use work.dp32_types.all;

entity cond_code_comparator is
generic (Tpd : Time := unit_delay);
port (cc : in CC_bits;

cm : in cm_bits;
result : out bit);

end cond_code_comparator;

architecture behaviour of cond_code_comparator is
alias cc_V : bit is cc(2);
alias cc_N : bit is cc(1);
alias cc_Z : bit is cc(0);
alias cm_i : bit is cm(3);
alias cm_V : bit is cm(2);
alias cm_N : bit is cm(1);
alias cm_Z : bit is cm(0);

begin
result <= bool_to_bit(((cm_V and cc_V)

or (cm_N and cc_N)
or (cm_Z and cc_Z)) = cm_i) after Tpd;

end behaviour;

use work.dp32_types.all, work.ALU_32_types.all;

architecture RTL of dp32 is

component reg_file_32_rrw
generic (depth : positive);
port (a1 : in bit_vector(depth-1 downto 0);

q1 : out bus_bit_32 bus;
en1 : in bit;
a2 : in bit_vector(depth-1 downto 0);
q2 : out bus_bit_32 bus;
en2 : in bit;
a3 : in bit_vector(depth-1 downto 0);
d3 : in bit_32;
en3 : in bit);

end component;

component mux2
generic (width : positive);
port (i0, i1 : in bit_vector(width-1 downto 0);

y : out bit_vector(width-1 downto 0);
sel : in bit);

end component;

component PC_reg
port (d : in bit_32;

q : out bus_bit_32 bus;
latch_en : in bit;
out_en : in bit;
reset : in bit);

end component;

component ALU_32
port (operand1 : in bit_32;

operand2 : in bit_32;
result : out bus_bit_32 bus;
cond_code : out CC_bits;
command : in ALU_command);

end component;

component cond_code_comparator
port (cc : in CC_bits;

cm : in cm_bits;
result : out bit);

end component;

component buffer_32
port (a : in bit_32;

b : out bus_bit_32 bus;
en : in bit);

end component;

component latch
generic (width : positive);
port (d : in bit_vector(width-1 downto 0);

q : out bit_vector(width-1 downto 0);
en : in bit);

end component;

component latch_buffer_32
port (d : in bit_32;

q : out bus_bit_32 bus;
latch_en : in bit;
out_en : in bit);

end component;

component signext_8_32
port (a : in bit_8;

b : out bus_bit_32 bus;
en : in bit);

end component;

signal op1_bus : bus_bit_32;
signal op2_bus : bus_bit_32;
signal r_bus : bus_bit_32;

signal ALU_CC : CC_bits;
signal CC : CC_bits;

signal current_instr : bit_32;
alias instr_a1 : bit_8 is current_instr(15 downto 8);
alias instr_a2 : bit_8 is current_instr(7 downto 0);
alias instr_a3 : bit_8 is current_instr(23 downto 16);
alias instr_op : bit_8 is current_instr(31 downto 24);
alias instr_cm : cm_bits is current_instr(19 downto 16);

signal reg_a2 : bit_8;
signal reg_result : bit_32;

signal addr_latch_en : bit;
signal disp_latch_en : bit;
signal disp_out_en : bit;
signal d2_en : bit;
signal dr_en : bit;
signal instr_latch_en : bit;
signal immed_signext_en : bit;
signal ALU_op : ALU_command;
signal CC_latch_en : bit;
signal CC_comp_result : bit;
signal PC_latch_en : bit;
signal PC_out_en : bit;
signal reg_port1_en : bit;
signal reg_port2_en : bit;
signal reg_port3_en : bit;
signal reg_port2_mux_sel : bit;
signal reg_res_latch_en : bit;

begin -- architecture RTL of dp32

reg_file : reg_file_32_RRW
generic map (depth => 8)
port map (a1 => instr_a1, q1 => op1_bus, en1 => reg_port1_en,

a2 => reg_a2, q2 => op2_bus, en2 => reg_port2_en,
a3 => instr_a3, d3 => reg_result, en3 => reg_port3_en);

reg_port2_mux : mux2
generic map (width => 8)
port map (i0 => instr_a2, i1 => instr_a3, y => reg_a2,

sel => reg_port2_mux_sel);

reg_res_latch : latch
generic map (width => 32)
port map (d => r_bus, q => reg_result, en => reg_res_latch_en);

PC : PC_reg
port map (d => r_bus, q => op1_bus,

latch_en => PC_latch_en, out_en => PC_out_en,
reset => reset);

ALU : ALU_32
port map (operand1 => op1_bus, operand2 => op2_bus,

result => r_bus, cond_code => ALU_CC,
command => ALU_op);

CC_reg : latch
generic map (width => 3)
port map (d => ALU_CC, q => CC, en => CC_latch_en);

CC_comp : cond_code_comparator
port map (cc => CC, cm => instr_cm, result => CC_comp_result);

dr_buffer : buffer_32
port map (a => d_bus, b => r_bus, en => dr_en);

d2_buffer : buffer_32
port map (a => op2_bus, b => d_bus, en => d2_en);

disp_latch : latch_buffer_32
port map (d => d_bus, q => op2_bus,

latch_en => disp_latch_en, out_en => disp_out_en);

addr_latch : latch
generic map (width => 32)
port map (d => r_bus, q => a_bus, en => addr_latch_en);

instr_latch : latch
generic map (width => 32)
port map (d => r_bus, q => current_instr, en => instr_latch_en);

immed_signext : signext_8_32
port map (a => instr_a2, b => op2_bus, en => immed_signext_en);

controller : block

port (phi1, phi2 : in bit;
reset : in bit;
opcode : in bit_8;
read, write, fetch : out bit;
ready : in bit;
addr_latch_en : out bit;
disp_latch_en : out bit;
disp_out_en : out bit;
d2_en : out bit;
dr_en : out bit;
instr_latch_en : out bit;
immed_signext_en : out bit;
ALU_op : out ALU_command;
CC_latch_en : out bit;
CC_comp_result : in bit;
PC_latch_en : out bit;
PC_out_en : out bit;
reg_port1_en : out bit;
reg_port2_en : out bit;
reg_port3_en : out bit;
reg_port2_mux_sel : out bit;
reg_res_latch_en : out bit);

port map (phi1 => phi1, phi2 => phi2,
reset => reset,
opcode => instr_op,
read => read, write => write, fetch => fetch,
ready => ready,
addr_latch_en => addr_latch_en,
disp_latch_en => disp_latch_en,
disp_out_en => disp_out_en,
d2_en => d2_en,
dr_en => dr_en,
instr_latch_en => instr_latch_en,
immed_signext_en => immed_signext_en,
ALU_op => ALU_op,
CC_latch_en => CC_latch_en,
CC_comp_result => CC_comp_result,
PC_latch_en => PC_latch_en, PC_out_en => PC_out_en,
reg_port1_en => reg_port1_en,
reg_port2_en => reg_port2_en,
reg_port3_en => reg_port3_en,
reg_port2_mux_sel => reg_port2_mux_sel,
reg_res_latch_en => reg_res_latch_en);

begin -- block controller

state_machine: process

type controller_state is
(resetting, fetch_0, fetch_1, fetch_2, decode,

disp_fetch_0, disp_fetch_1, disp_fetch_2,
execute_0, execute_1, execute_2);

variable state, next_state : controller_state;
variable write_back_pending : boolean;

type ALU_op_select_table is
array (natural range 0 to 255) of ALU_command;

constant ALU_op_select : ALU_op_select_table :=
(16#00# => add,

16#01# => subtract,
16#02# => multiply,
16#03# => divide,
16#10# => add,
16#11# => subtract,
16#12# => multiply,
16#13# => divide,
16#04# => log_and,
16#05# => log_or,
16#06# => log_xor,
16#07# => log_mask,
others => disable);

begin -- process state_machine
--
-- start of clock cycle
--
wait until phi1 = '1';
--
-- check for reset
--
if reset = '1' then

state := resetting;
--
-- reset external bus signals
--
read <= '0' after Tpd;
fetch <= '0' after Tpd;
write <= '0' after Tpd;
--
-- reset dp32 internal control signals
--
addr_latch_en <= '0' after Tpd;
disp_latch_en <= '0' after Tpd;
disp_out_en <= '0' after Tpd;
d2_en <= '0' after Tpd;
dr_en <= '0' after Tpd;
instr_latch_en <= '0' after Tpd;
immed_signext_en <= '0' after Tpd;
ALU_op <= disable after Tpd;
CC_latch_en <= '0' after Tpd;
PC_latch_en <= '0' after Tpd;
PC_out_en <= '0' after Tpd;
reg_port1_en <= '0' after Tpd;
reg_port2_en <= '0' after Tpd;
reg_port3_en <= '0' after Tpd;
reg_port2_mux_sel <= '0' after Tpd;
reg_res_latch_en <= '0' after Tpd;
--
-- clear write-back flag
--
write_back_pending := false;
--

else -- reset = '0'
state := next_state;

end if;

--
-- dispatch action for current state
--
case state is

when resetting =>
--
-- check for reset going inactive at end of clock cycle
--
wait until phi2 = '0';
if reset = '0' then

next_state := fetch_0;
else

next_state := resetting;
end if;
--

when fetch_0 =>
--
-- clean up after previous execute cycles
--
reg_port1_en <= '0' after Tpd;
reg_port2_mux_sel <= '0' after Tpd;
reg_port2_en <= '0' after Tpd;
immed_signext_en <= '0' after Tpd;
disp_out_en <= '0' after Tpd;
dr_en <= '0' after Tpd;
read <= '0' after Tpd;
d2_en <= '0' after Tpd;
write <= '0' after Tpd;
--
-- handle pending register write-back
--
if write_back_pending then

reg_port3_en <= '1' after Tpd;
end if;
--
-- enable PC via ALU to address latch
--
PC_out_en <= '1' after Tpd; -- enable PC onto op1_bus
ALU_op <= pass1 after Tpd; -- pass PC to r_bus
--
wait until phi2 = '1';
addr_latch_en <= '1' after Tpd; -- latch instr address
wait until phi2 = '0';
addr_latch_en <= '0' after Tpd;
--
next_state := fetch_1;
--

when fetch_1 =>
--
-- clear pending register write-back
--
if write_back_pending then

reg_port3_en <= '0' after Tpd;
write_back_pending := false;

end if;
--
-- increment PC & start bus read
--
ALU_op <= incr1 after Tpd; -- increment PC onto r_bus
fetch <= '1' after Tpd;
read <= '1' after Tpd;
--
wait until phi2 = '1';
PC_latch_en <= '1' after Tpd; -- latch incremented PC
wait until phi2 = '0';
PC_latch_en <= '0' after Tpd;
--
next_state := fetch_2;
--

when fetch_2 =>
--
-- cleanup after previous fetch_1
--
PC_out_en <= '0' after Tpd; -- disable PC from op1_bus
ALU_op <= disable after Tpd; -- disable ALU from r_bus

--
-- latch current instruction
--
dr_en <= '1' after Tpd; -- enable fetched instr onto r_bus
--
wait until phi2 = '1';
instr_latch_en <= '1' after Tpd;  -- latch fetched instr from r_bus
wait until phi2 = '0';
instr_latch_en <= '0' after Tpd;
--
if ready = '1' then

next_state := decode;
else

next_state := fetch_2; -- extend bus read
end if;

when decode =>
--
-- terminate bus read from previous fetch_2
--
fetch <= '0' after Tpd;
read <= '0' after Tpd;
dr_en <= '0' after Tpd; -- disable fetched instr from r_bus
--
-- delay to allow decode logic to settle
--
wait until phi2 = '0';
--
-- next state based on opcode of currect instruction
--
case opcode is

when op_add | op_sub | op_mul | op_div
| op_addq | op_subq | op_mulq | op_divq
| op_land | op_lor | op_lxor | op_lmask
| op_ldq | op_stq =>

next_state := execute_0;
when op_ld | op_st =>

next_state := disp_fetch_0; -- fetch offset
when op_br | op_bi =>

if CC_comp_result = '1' then -- if branch taken
next_state := disp_fetch_0; -- fetch displacement

else -- else
next_state := execute_0; -- increment PC

-- past displacement
end if;

when op_brq | op_biq =>
if CC_comp_result = '1' then -- if branch taken

next_state := execute_0; -- add immed
-- displacement to PC

else -- else
next_state := fetch_0; -- no action needed

end if;
when others =>

assert false report "illegal instruction" severity warning;
next_state := fetch_0; -- ignore and carry on

end case; -- op
--

when disp_fetch_0 =>
--
-- enable PC via ALU to address latch
--
PC_out_en <= '1' after Tpd; -- enable PC onto op1_bus
ALU_op <= pass1 after Tpd; -- pass PC to r_bus
--
wait until phi2 = '1';
addr_latch_en <= '1' after Tpd; -- latch displacement address
wait until phi2 = '0';
addr_latch_en <= '0' after Tpd;
--
next_state := disp_fetch_1;
--

when disp_fetch_1 =>
--
-- increment PC & start bus read
--
ALU_op <= incr1 after Tpd; -- increment PC onto r_bus
fetch <= '1' after Tpd;
read <= '1' after Tpd;
--
wait until phi2 = '1';
PC_latch_en <= '1' after Tpd; -- latch incremented PC
wait until phi2 = '0';
PC_latch_en <= '0' after Tpd;
--
next_state := disp_fetch_2;
--

when disp_fetch_2 =>
--
-- cleanup after previous disp_fetch_1
--
PC_out_en <= '0' after Tpd; -- disable PC from op1_bus
ALU_op <= disable after Tpd; -- disable ALU from r_bus
--
-- latch displacement
--
wait until phi2 = '1';
disp_latch_en <= '1' after Tpd;  -- latch fetched disp from r_bus
wait until phi2 = '0';
disp_latch_en <= '0' after Tpd;
--
if ready = '1' then

next_state := execute_0;
else

next_state := disp_fetch_2; -- extend bus read
end if;

when execute_0 =>
--
-- terminate bus read from previous disp_fetch_2
--
fetch <= '0' after Tpd;
read <= '0' after Tpd;
--
case opcode is

when op_add | op_sub | op_mul | op_div
| op_addq | op_subq | op_mulq | op_divq
| op_land | op_lor | op_lxor | op_lmask =>

-- enable r1 onto op1_bus
reg_port1_en <= '1' after Tpd;
if opcode = op_addq or opcode = op_subq

or opcode = op_mulq or opcode = op_divq then
-- enable i8 onto op2_bus
immed_signext_en <= '1' after Tpd;

else
-- select a2 as port2 address
reg_port2_mux_sel <= '0' after Tpd;
-- enable r2 onto op2_bus
reg_port2_en <= '1' after Tpd;

end if;
-- select ALU operation
ALU_op <= ALU_op_select(bits_to_int(opcode)) after Tpd;
--
wait until phi2 = '1';
-- latch cond codes from ALU
CC_latch_en <= '1' after Tpd;
-- latch result for reg write
reg_res_latch_en <= '1' after Tpd;
wait until phi2 = '0';
CC_latch_en <= '0' after Tpd;
reg_res_latch_en <= '0' after Tpd;
--
next_state := fetch_0; -- execution complete
write_back_pending := true; -- register write_back required
--

when op_ld | op_st | op_ldq | op_stq =>
-- enable r1 to op1_bus
reg_port1_en <= '1' after Tpd;
if opcode = op_ld or opcode = op_st then

-- enable displacement to op2_bus
disp_out_en <= '1' after Tpd;

else
-- enable i8 to op2_bus
immed_signext_en <= '1' after Tpd;

end if;
ALU_op <= add after Tpd; -- effective address to r_bus
--
wait until phi2 = '1';
addr_latch_en <= '1' after Tpd; -- latch effective address
wait until phi2 = '0';
addr_latch_en <= '0' after Tpd;
--
next_state := execute_1;
--

when op_br | op_bi | op_brq | op_biq =>
if CC_comp_result = '1' then

if opcode = op_br then
PC_out_en <= '1' after Tpd;
disp_out_en <= '1' after Tpd;

elsif opcode = op_bi then
reg_port1_en <= '1' after Tpd;
disp_out_en <= '1' after Tpd;

elsif opcode = op_brq then
PC_out_en <= '1' after Tpd;
immed_signext_en <= '1' after Tpd;

else -- opcode = op_biq
reg_port1_en <= '1' after Tpd;
immed_signext_en <= '1' after Tpd;

end if;
ALU_op <= add after Tpd;

else
assert opcode = op_br or opcode = op_bi

report "reached state execute_0 "
& "when brq or biq not taken"

severity error;
PC_out_en <= '1' after Tpd;
ALU_op <= incr1 after Tpd;

end if;
--
wait until phi2 = '1';
PC_latch_en <= '1' after Tpd; -- latch incremented PC
wait until phi2 = '0';
PC_latch_en <= '0' after Tpd;
--
next_state := fetch_0;
--

when others =>
null;

end case; -- op
--

when execute_1 =>
--
-- opcode is load or store instruction.
-- cleanup after previous execute_0
--
reg_port1_en <= '0' after Tpd;
if opcode = op_ld or opcode = op_st then

-- disable displacement from op2_bus
disp_out_en <= '0' after Tpd;

else
-- disable i8 from op2_bus
immed_signext_en <= '0' after Tpd;

end if;
ALU_op <= add after Tpd; -- disable ALU from r_bus
--
-- start bus cycle
--
if opcode = op_ld or opcode = op_ldq then

fetch <= '0' after Tpd; -- start bus read
read <= '1' after Tpd;

else -- opcode = op_st or opcode = op_stq
reg_port2_mux_sel <= '1' after Tpd; -- address a3 to port2
reg_port2_en <= '1' after Tpd; -- reg port2 to op2_bus
d2_en <= '1' after Tpd; -- enable op2_bus to d_bus buffer
write <= '1' after Tpd; -- start bus write

end if;
--
next_state := execute_2;
--

when execute_2 =>
--
-- opcode is load or store instruction.
-- for load, enable read data onto r_bus
--
if opcode = op_ld or opcode = op_ldq then

dr_en <= '1' after Tpd; -- enable data to r_bus
wait until phi2 = '1';
-- latch data in reg result latch
reg_res_latch_en <= '1' after Tpd;
wait until phi2 = '0';
reg_res_latch_en <= '0' after Tpd;
write_back_pending := true; -- write-back pending

end if;
--
next_state := fetch_0;
--

end case; -- state
end process state_machine;

end block controller;

end RTL;

phi1

phi2

reg_port1_en

ALU_op

addr_latch_en

a_bus

d_bus

ready

read

execute_0

add

disp_out_en
or  immed_
signext_en

store address

disable

valid data out

reg_port2_
mux_sel

reg_port2_en

d2_en

fetch

write

execute_1 execute_2

use work.dp32_types.all;

configuration dp32_rtl_test of dp32_test is

for structure
for cg : clock_gen

use entity work.clock_gen(behaviour)
generic map (Tpw => 8 ns, Tps => 2 ns);

end for;
for mem : memory

use entity work.memory(behaviour);
end for;
for proc : dp32

use entity work.dp32(rtl);
for rtl

for all : reg_file_32_rrw
use entity work.reg_file_32_rrw(behaviour);

end for;
for all : mux2

use entity work.mux2(behaviour);
end for;
for all : latch

use entity work.latch(behaviour);
end for;
for all : PC_reg

use entity work.PC_reg(behaviour);
end for;
for all : ALU_32

use entity work.ALU_32(behaviour);
end for;
for all : cond_code_comparator

use entity work.cond_code_comparator(behaviour);
end for;
for all : buffer_32

use entity work.buffer_32(behaviour);
end for;
for all : latch_buffer_32

use entity work.latch_buffer_32(behaviour);
end for;
for all : signext_8_32

use entity work.signext_8_32(behaviour);
end for;

end for;
end for;

end for;

end dp32_rtl_test;

